

## 21. Sequence, Set, and List

A set is a sequence enclosed by curly braces, a list is a sequence enclosed by square braces. There are two more fundamental differences between sets and lists. First, every element in a set occurs only once, while elements may repeat themselves in a list. Second, in a set Maple may arrange the order of the elements as it pleases, while the order of the elements in a list stays fixed.

### 21.1 Sequence

We have seen how to build sequences with higher order derivatives:

```
[> sqx24 := x$24;  
[> whattype(sqx24);
```

To make a sequence of indexed elements we use the seq command:

```
[> indseq := seq(x[i], i=1..24);
```

Observe the difference with a sequence where the indices are concatenated to x:

```
[> cctseq := seq(x||i, i=1..24);
```

We know already how to get the number of elements in a sequence and how to select from a sequence:

```
[> indseq[13] , "is not the same as " , cctseq[13];
```

To get the number of elements in a sequence, we need to convert to a list:

```
[> nops([indseq]);
```

Suppose we wish to make the sequence a1,a2,...,b1,b2,...,c1,c2,...:

```
[> a||(1..5);  
[> (a,b,c)||(1..5);
```

The last command gives an error message. We have to concatenate to the empty symbol “ (two left quotes):

```
[> ‘||(a,b,c)||(1..5);
```

### 21.2 Set

A set in Maple corresponds to the mathematical notion of a set: there is no order and the set contains no duplicates.

```
[> S := {a,b,c};  
[> whattype(S);  
[> nops(S);  
[> op(2,S);  
[> S2 := {b,d,e};  
[> S union S2; S intersect S2;
```

We can build sets of sets, for example the set of all subsets:

```
[> pS := combinat[powerset](S);
```

The number of elements in the set of all subsets is  $2^{nops(S)}$ . This cardinality (two to the power the number of elements in the set) helps understanding the naming of the command.

If we want repeated elements, we need to use a list:

```
[> L := [a, a, b];
[> pL := combinat[powerset](L);
```

Because of the repetition, we have fewer than 8 elements in pL.

Note that we can easily remove duplicates from a list by converting to a set and back :

```
[> LL := [op({op(L)})];
```

So the conversion from list to set and set to list went via a sequence (twice used op).

### 21.3 List

Remember the anonymous functions we have seen to execute on a list. Let us make a list of 20 random numbers between 1 and 6:

```
[> die := rand(1..6);
```

Here is the short way to build a list :

```
[> ranlist := [die()$20];
```

What went wrong? We must prevent premature evaluation:

```
[> randlist := ['die()'$20];
```

Here is an alternative way (in case you forgot about the right quotes) :

```
[> secondlist := [];
[> for i from 1 to 20 do
[>   secondlist := [op(secondlist),die()]:
[> end do:
[> secondlist;
```

It is interesting to replace the colons in the loop above by semicolons to see the evolution of the build up. Let us now make a list of random pairs:

```
[> z1 := zip((x,y) -> [x,y],randlist,secondlist);
```

And here we show how to “flatten” the list :

```
[> map(op,z1);
```

### 21.4 Assignments

- Use the command **ithprime** to generate a sequence of the first 100 prime numbers. To visualize how this sequence grows, do the following:
  - Convert the sequence of the 100 first prime numbers to a list;
  - Generate a list of the first 100 natural numbers (starting at 1);
  - Combine the two lists into a list of points, like  $[[1, 2], [2, 3], [3, 5], [4, 7], [5, 11], [6, 13], \dots]$ , so the  $i$ th point in this list is  $[i, \text{ithprime}(i)]$ ;
  - Calling the list of points `l`, type `plots[pointplot](l)`; to visualize the sequence of primes.
- Consider `f := n -> trunc(evalf(Pi*10^n, n+3)) mod 10`;  
 $f(n)$  returns the  $n$ th digit after the decimal point of  $\pi$ , i.e.:  $f(0) = 3, f(1) = 1, f(2) = 4, \dots$ 
  - Use `f` to generate the list of the first 100 significant decimal places of  $\pi$ . Call this list `l`.
  - Given `l`, what is the Maple command to create an approximation for  $\pi$  with 100 decimal places? Compare your answer to `evalf(Pi, 100)`.